

## EU-AES Tools: Implementation of the European Socioeconomic Groups Classification (ESeG) using Adult Education Survey Microdata

Bohr, Jeanette

Veröffentlichungsversion / Published Version  
Arbeitspapier / working paper

**Zur Verfügung gestellt in Kooperation mit / provided in cooperation with:**  
GESIS - Leibniz-Institut für Sozialwissenschaften

### Empfohlene Zitierung / Suggested Citation:

Bohr, J. (2018). *EU-AES Tools: Implementation of the European Socioeconomic Groups Classification (ESeG) using Adult Education Survey Microdata*. (GESIS Papers, 2018/14). Köln: GESIS - Leibniz-Institut für Sozialwissenschaften.  
<https://doi.org/10.21241/ssoar.57622>

### Nutzungsbedingungen:

Dieser Text wird unter einer CC BY-NC Lizenz (Namensnennung-Nicht-kommerziell) zur Verfügung gestellt. Nähere Auskünfte zu den CC-Lizenzen finden Sie hier:  
<https://creativecommons.org/licenses/by-nc/4.0/deed.de>

### Terms of use:

This document is made available under a CC BY-NC Licence (Attribution-NonCommercial). For more Information see:  
<https://creativecommons.org/licenses/by-nc/4.0>

## EU-AES Tools:

Implementation of the European  
Socioeconomic Groups Classification  
(ESeG) using Adult Education Survey  
Microdata

*Jeanette Bohr*



GESIS Papers 2018|14

## **EU-AES Tools:**

**Implementation of the European  
Socioeconomic Groups Classification (ESeG)  
using Adult Education Survey Microdata**

*Jeanette Bohr*

## **GESIS Papers**

GESIS – Leibniz-Institut für Sozialwissenschaften  
Dauerbeobachtung der Gesellschaft  
German Microdata Lab  
Postfach 12 21 55  
68072 Mannheim  
E-Mail: [jeanette.bohr@gesis.org](mailto:jeanette.bohr@gesis.org)

ISSN:	2364-3781 (Online)
Herausgeber,	
Druck und Vertrieb:	GESIS – Leibniz-Institut für Sozialwissenschaften Unter Sachsenhausen 6-8, 50667 Köln

## Summary

---

This GESIS Paper describes the implementation of the ESeG (European socioeconomic groups) Classification with anonymised microdata of the Adult Education Survey (AES) 2011. This classification is designed to measure socio-economic status at European level. ESeG is an update of the European Socio-Economic Classification (ESeC) initiated by Eurostat. The update became necessary due to the revision of the ISCO occupational classification in 2008. The ESeG classification aims to provide a harmonized and validated classification that can be implemented easily with data from European statistics and thus supports comparative research in Europe.

## Zusammenfassung

---

Dieses GESIS Paper beschreibt die Umsetzung der ESeG (European socioeconomic groups) Klassifikation mit den anonymisierten Mikrodaten des Adult Education Survey (AES) 2011. Die Klassifikation dient der Messung des sozioökonomischen Status auf europäischer Ebene. ESeG ist eine Aktualisierung der von Eurostat initiierten Europäischen Sozioökonomischen Klassifikation ESeC. Die Aktualisierung wurde durch die Überarbeitung der ISCO-Berufsklassifikation im Jahr 2008 notwendig. Ziel der ESeG-Klassifikation ist es, eine harmonisierte und validierte Klassifikation bereitzustellen, die leicht mit Daten der europäischen Statistik umgesetzt werden kann und damit die vergleichende Forschung in Europa unterstützt.

## 1 Introduction

---

The German Microdata Lab (GML) already offers some tools and services for the use of ESeC and ESeG.<sup>1</sup> The microdata tool described here complements this service for users of European microdata and provides an implementation for the Adult Education Survey (AES) 2001 data.

### 1.1 The ESeG Project

The socioeconomic position is a relevant explanatory variable in many sociological research projects. These projects face the challenge that standardized and internationally harmonised indicators are needed to measure the individual economic and social position in comparison to others. In the early 2000s, a harmonised instrument for measuring the socioeconomic situation of the European region was developed: the European socioeconomic classification ESeC<sup>2</sup>. The theoretical model followed the Erikson-Goldthorpe class scheme (EGP) (Erikson/Goldthorpe 1992). In ESeC, the class assignment depends on the position in working life and the type of employment relationship.

With the revision of the ISCO occupational classification in 2008, a revision of the ESeC classification became necessary, as ESeC is based on ISCO-88 and uses a definition of the supervisory function that does not correspond to that used in ISCO-08. To this end, Eurostat has launched the project "ESSnet ESEG"<sup>3</sup> (October 2011 to April 2014) to develop and test a new classification – the European socioeconomic groups (ESeG). The starting point was the experience from the ESeC<sup>4</sup> project, but the new classification is based on ISCO-08 (see Meron et al. 2014).

The project aimed to develop a classification that can be easily implemented with data from the European Union. During the development of the classification, three prototypes were tested, of which one was finally chosen. The tests consisted of various kinds of variance analyses using microdata from Eurostat's most frequently used European surveys, including EU-SILC, EU-LFS, and AES. The aim was to find out which prototype has the highest variance between the ESeG groups and the smallest variance within a single group. The selected prototype is the one that fits best economic and labour market issues and is most suitable for examining inequalities in European countries (see Meron et al. 2014: 27ff.).

The ESeG classification only requires the ISCO 08 job classification and two core variables for employment status and information on the unemployed. ESeG is suitable for studies at the European and the national level and can be used for transnational comparisons.

---

<sup>1</sup> Syntax to generate ESeC and ESeG using EU-SILC SPSS or Stata Systemfiles:  
<https://www.gesis.org/en/gml/european-microdata/eu-silc/>

Syntax to generate ESeC using German Microcensus SPSS or Stata Systemfiles:  
<https://www.gesis.org/en/missy/materials/MZ/tools/esec>

<sup>2</sup> For details on ESeC please visit the homepage of Eric Harrison:  
<https://ekharrison.weebly.com/european-socio-economic-classification-esec.html>

<sup>3</sup> For details on ESeG project please visit the ESSnet project website:  
[https://ec.europa.eu/eurostat/cros/content/eseg\\_en](https://ec.europa.eu/eurostat/cros/content/eseg_en)

<sup>4</sup> See <https://www.iser.essex.ac.uk/archives/esec>

## 1.2 Dataset: Adult Education Survey

The Adult Education Survey (AES) covers the participation of adults in education and training and is one of the main data sources for statistics on lifelong learning in the EU. The survey focuses on people aged 25-64 who live in private households. The AES provides information on adult participation in formal education, non-formal education and training and informal learning in Europe. The survey contents are self-reported competencies, participation in learning and educational activities, characteristics of learning activities, participation in cultural and social activities and socio-demographic background information.

The survey is conducted every five years. Following the first pilot survey (2005-2008), a legal basis was established for the Adult Education Survey: Framework Regulation (EC) 452/2008 (European Commission 2013). The second AES data collection took place between July 2011 and June 2012. The third AES survey was conducted in 2016 and 2017.

The current legal framework provides access to anonymised microdata available at Eurostat for scientific purposes. So far the first wave (AES 2007) and the second wave (AES 2011) are available as scientific use files.<sup>5</sup>

Structured metadata on AES and setup routines for converting Eurostat data files (csv) to SPSS system files are available online in the Microdata Information System MISSY<sup>6</sup>.

---

<sup>5</sup> <http://ec.europa.eu/eurostat/web/microdata/adult-education-survey>

<sup>6</sup> <https://www.gesis.org/en/missy/metadata/AES/>



## 2 ESeG coding scheme

---

ESeG is a two-level classification consisting of nine main groups and 42 subgroups. The nine main groups are divided into seven categories for the distinction of economically active and two categories for inactive persons. The classification, therefore, represents all persons in the sample.

ESeG categories 1–7 combine information on the occupation and status of respondents in employment. The categories can be divided into three classes: high class (1+2), middle class (3+4) and working class (5+6+7). The groups are broken down into several subgroups.

ESeG category 8 covers pensioners and inactive persons aged 65 and over. Originally, the category was divided into 8 subcategories, based on the employment status in the last job. However, since the information about employment in the last job is missing, these subgroups cannot be generated with AES data (see chapter 4.2.2)

ESeG category 9 covers other non-active persons. The category is divided into four groups: students, permanently disabled, unemployed persons not classified elsewhere and other inactive persons younger than 65 years of age.

### **The European Socio-economic Groups (ESEG) coding scheme according Meron et al. (2014): groups and sub-groups**

#### **1 Managers (all statuses)**

- 1.1 Higher managerial self-employed (ISCO 11, 12, 13 and status=se)
- 1.2 Lower managerial self-employed (ISCO 14 and status=se)
- 1.3 Higher managerial employees (ISCO 11, 12, 13 and 01 and status=e)
- 1.4 Lower managerial employees (ISCO 14 and status=e)

#### **2 Professionals (all statuses)**

- 2.1 Science, engineering and information and communications technology (ICT) professionals (ISCO 21, 25)
- 2.2 Health professionals (ISCO 22)
- 2.3 Business and administration professionals (ISCO 24)
- 2.4 Legal, social and cultural professionals (ISCO 26)
- 2.5 Teaching professionals (ISCO 23)

#### **3 Technicians and associated professionals employees (status=e)**

- 3.1 Science, engineering and ICT technicians and associated professionals (ISCO 31, 35)
- 3.2 Health associate professionals (ISCO 32)
- 3.3 Business and administration associate professionals (ISCO 33)
- 3.4 Legal, social and cultural associate professionals (ISCO 34)
- 3.5 Non-commissioned armed forces officers (ISCO 02)

#### **4 Small entrepreneurs (status=se)**

- 4.1 Self-employed agricultural and related workers' (ISCO 6)
- 4.2 Self-employed technicians, clerical support, services and sales workers (ISCO 3, 4, 5)
- 4.3 Self-employed drivers, craft, trades and elementary workers (ISCO 7, 8, 9)

#### **5 Clerks and skilled service employees (status=e)**

- 5.1 General and numerical clerks and other clerical support employees (ISCO 41, 43, 44)
- 5.2 Customer services clerks (ISCO 42)
- 5.3 Personal care employees (ISCO 53)
- 5.4 Protective service employees and armed forces, other ranks (ISCO 03 and ISCO 54)

**6 Skilled industrial employees (status=e)**

- 6.1 Building and related trade employees (ISCO 71)
- 6.2 Food processing, wood working, garment employees (ISCO 75)
- 6.3 Metal, machinery, handicraft, printing, electrical and electronic trades employees (ISCO 72, 73, 74)
- 6.4 Stationary plant and machinery operation and assembly employees (ISCO 81, 82)
- 6.5 Employee drivers and mobile plant operators (ISCO 83)

**7 Lower status employees (status=e)**

- 7.1 Personal services and sales employees (ISCO 51, 52)
- 7.2 Blue collar employees and food preparation assistants in elementary occupations (ISCO 92, 93, 94, 96)
- 7.3 Cleaners and helpers and services employees in elementary occupations (ISCO 91, 95)
- 7.4 Agricultural employees (ISCO 6)

**8 Retired persons (and people 65 and over non-employed)**

- 8.1 Retired Managers
- 8.2 Retired professionals
- 8.3 Retired technicians and associated professionals employees
- 8.4 Retired small entrepreneurs
- 8.5 Retired clerks and skilled service employees
- 8.6 Retired skilled Industrial employees
- 8.7 Retired Lower status employees
- 8.8 Other persons outside the labour force aged 65 or more

**9 Other non-employed persons**

- 9.1 Students
- 9.2 Permanently disabled
- 9.3 Unemployed not elsewhere classified
- 9.4 Other persons outside the labour force aged less than 65 years

## 3 Routine Description

---

### 3.1 Variables Required

Four variables are required to measure ESeG, in particular, the two core variables ISCO-08 occupation and employment status (employees/self-employed). For persons without gainful employment, information on status (pensioners/students/disabled persons) and age is required.

#### 3.1.1 ISCO-08

The variable JOBICO ("Occupation") of the AES 2011 covers the profession of the interviewee according to the ISCO 08 classification. The reference period is the time of the interview. It is important to note that the variable in the AES data refers to the main activity and is only available for people currently employed.

#### 3.1.2 Employment Status

Information on employment status is stored in the AES 2011 variable JOBSTAT ("Professional status"). The variable covers persons currently employed and distinguishes between self-employed persons with and without employees, employees with permanent and fixed-term contracts and family workers.

#### 3.1.3 Main current labour status

The variable MAINSTAT ("Main current labour status") of the AES 2011 records the main activity at the time of the interview. This variable is used to classify whether respondents are employed, unemployed or inactive. The categories also distinguish between several groups of inactive persons (e.g. retirement or permanent disability).

#### 3.1.4 Age

The variable AGE ("Age") is used to define the working age population (25–64 years) and the group of non-employed persons aged 65 and over (for ESeG category 8 "Retired persons"). In most countries, permanent residents aged 25–64 are interviewed. In addition, some countries also cover the age groups 18 to 24 years and 65 to 75 years. The ESeG implementation includes all available age groups. To compare countries, however, it is necessary to select people between the ages of 25 and 64. Therefore, a second variable is generated in the SPSS routine, which only refers to the working age population.

### 3.2 AES specifics

#### 3.2.1 Country specifics

Implementation of the ESeG is not possible for two countries: Malta and Ireland. For Malta, ISCO-08 is only offered for the main groups (1-digit). Data for Ireland include implausible values in the variable MAINSTAT (53.4% Item Nonresponse). In the syntax routine, both countries are excluded from implementation.

For Hungary, the categories of inactive persons are grouped into a single category (MAINSTAT=30). It is therefore not possible to distinguish correctly between ESeG groups 8 and 9. Since only inactive persons under 65 are covered for Hungary, all inactive persons are classified in ESeG category 9 (inactive and under 65 years).

### 3.2.2 Non-employed people

For the implementation of ESeG, it is recommended to classify non-working persons according to their last activity. If this information is not available, it should be classified in ESeG category 9.3 ("Unemployed not elsewhere classified") (Meron et al. 2014: 31). In the AES 2011 data, information about the respondent's occupation is only available to employed persons. ESeG categories 1 to 7 can therefore only be implemented for the group of employed persons. Moreover, the former socio-economic position of pensioners (ESeG subcategories 8.1 to 8.8) cannot be realised either.

### 3.2.3 Retired persons

ESeG category 8 ("Retired persons") is intended to cover both retired persons and non-employed persons aged 65 and over. As only a few countries in the AES survey the optional age group of 65 to 75 year-olds, only retired persons under 65 are included in ESeG category 8 in most countries. When comparing countries with different age limits, it is therefore necessary to restrict the calculation to persons less than 64 years of age.

## 3.3 Variable Preparation

First, some variables necessary for the calculation of ESeG must be recoded: (1) undo the definition of missing values in the variables MAINSTAT, JOBSTAT and JOBISCO, (2) recode the variable JOBISCO into a numeric variable, (3) define data from Malta and Ireland as MISSING in the ISCO variable to exclude both countries from the ESeG implementation, (4) convert the variable JOBSTAT into a binary variable STA (self-employed vs. employees).

## 3.4 ESeG level 1 (groups)

The variable ESEG\_1 defines 9 ESeG groups (aggregated level).

Categories 1 to 7 refer to persons currently employed (respondents with "Main current labour status" (MAINSTAT): full-time or part-time): (1) "Managers", (2) "Professionals", (3) "Technicians and associated professional employees", (4) "Small entrepreneurs", (5) "Clerks and skilled service employees", (6) "Skilled industrial employees", (7) "Lower status employees". The calculation of categories 1 to 7 combines information on the respondents' occupation (ISCO 08) and their status in employment (employed or self-employed). The assignment follows the ESeG scheme proposed by Meron et al. 2014.

Category 8 ("Retired persons") covers both pensioners and inactive persons aged 65 and over. It must be noted here that the information about inactive persons aged 65 and over is only available for a few countries.

Category 9 ("Other non-employed persons") includes inactive persons under 65 years of age and unemployed persons.

### 3.5 ESeG level 1 (groups), restricted to persons aged 25 to 64

The variable ESEGwa\_1 (ESeG level 1, working age) defines 9 ESeG groups as in the variable ESEG1, but only includes persons aged 25 to 64. The samples from the individual countries contain different age limits (see table 1). When comparing countries, it is necessary to limit the sample to the age group available for all countries (25 to 64). In this case, the variable ESEGwa\_1 should be used.

Table 1: Age range by country

	AT	BE	BG	CH	CY	CZ	DE	DK	EE	EL	ES	FI	FR	HU	IE
MIN	18	25	18	25	25	18	25	25	25	18	17	25	18	25	25
MAX	64	64	64	64	64	69	64	64	64	64	65	64	64	64	64
	IT	LT	LU	LV	MT	NL	NO	PL	PT	RO	RS	SE	SI	SK	UK
MIN	16	18	25	25	25	25	16	18	17	18	25	25	18	18	24
MAX	75	69	64	64	64	64	68	69	70	69	64	64	69	64	79

Source: EU\_AES\_2011\_MICRODATA\_V1.0.SAV, own calculation.

### 3.6 ESeG level 2 (subgroups)

ESeG level 2 describes the socio-economic situation at a more detailed level. For this purpose, the nine groups of ESeG level 1 are divided into sub-groups. As with ESeG Level 1, the implementation of the subgroups follows the ESeG coding scheme of Meron et al. (2014), with three exceptions: (1) As information on the last employment of inactive persons is missing, categories 1.1 to 7.4 refer only to persons in employment. (2) Moreover, the group of retired persons cannot be classified under subcategories 8.1 to 8.7 (socio-economic position in the last position). In contrast to the proposed classification, only the main category (8.0) is implemented. (3) Since for Hungary the categories of inactive persons are grouped in a single category, it is not possible to generate categories 9.1 to 9.4. An additional category 9.5 is therefore introduced for inactive persons in Hungary.

### 3.7 ESeG level 2 (subgroups), restricted to persons aged 25 to 64

The variable ESEGwa\_2 (ESeG level 2, working age) defines all ESeG subgroups as in the variable ESEG\_2, but only includes persons aged 25 to 64. The variable ESEGwa\_2 is intended to be used for country comparisons.

### 3.8 Missing Data

The classification is designed to include all persons in the sample. Due to item non-response and country-specific anonymisation rules, however, some cases cannot be considered in the implementation. This applies to the following cases: missing data due to item nonresponse ("no answer") in variables JOBISCO, MAINSTAT and JOBSTAT and excluded data from Malta and Ireland because of anonymization. There are also 42 unexplained missing value cases probably due to data errors. Overall, within the new variables ESEG\_1 and ESEG\_2, 8 percent of the AES 2011 sample are not coded as valid cases.

## 4 Descriptive information on ESeG (AES 2011)

Table 2: Level 1: AES 2011 ESEG\_1 and ESEGwa\_1 (working age: 25-64)

	ESEG_1			ESEGwa_1		
	N	%	Valid %	N	%	Valid %
1 Managers	8195	3,6	4	8065	3,6	4,4
2 Professionals	24384	10,8	11,8	23832	10,6	13,1
3 Technicians and associated professional employees	17448	7,7	8,4	16863	7,5	9,2
4 Small entrepreneurs	15643	6,9	7,5	14885	6,6	8,2
5 Clerks and skilled service employees	16165	7,2	7,8	15391	6,8	8,4
6 Skilled industrial employees	23562	10,5	11,4	22189	9,8	12,2
7 Lower status employees	21719	9,6	10,5	20118	8,9	11
8 Retired persons (and other inactive aged 65+)	26617	11,8	12,8	18794	8,3	10,3
9 Other non-employed persons aged < 65	53633	23,8	25,9	42406	18,8	23,2
<i>Valid Total</i>	<i>207366</i>	<i>92</i>	<i>100</i>	<i>182543</i>	<i>81</i>	<i>100</i>
MISSING						
System	17981	8		17879	7,9	
99 Age <25 or >64				24925	11,1	
<i>Total</i>	<i>225347</i>	<i>100</i>		<i>225347</i>	<i>100</i>	

Source: EU\_AES\_2011\_MICRODATA\_V1.0.SAV, own calculation, not weighted.

Table 3: Level 1: AES 2011 ESEGwa\_1 (working age: 25-64) by Country (%)

	AT	BE	BG	CH	CY	CZ	DE	DK	EE	EL	ES	FI	FR	HU
1 Managers	4,3	5,8	5,2	7,8	6,2	3,3	2,3	3,4	9,3	3,1	3,5	4,3	3,7	2,8
2 Professionals	13,9	15,7	10,5	21,2	14,4	10,2	13,3	16,4	16,6	9,3	10,9	20,0	13,2	10,5
3 Technicians and associated professional employees	12,6	11,3	4,6	14,1	8,9	10,1	14,9	19,4	9,3	4,0	9,4	11,8	13,1	6,0
4 Small entrepreneurs	6,2	4,7	5,3	7,8	7,5	9,7	6,0	0,0	2,9	18,8	8,3	7,6	6,3	6,0
5 Clerks and skilled service employees	9,3	13,0	7,3	9,8	8,8	9,1	9,7	7,3	5,9	6,6	6,9	9,1	10,7	7,3
6 Skilled industrial employees	11,5	8,7	15,9	10,2	9,9	18,2	10,8	10,3	18,2	6,9	10,0	11,3	9,7	15,8
7 Lower status employees	12,6	9,0	12,5	10,6	18,7	10,2	12,2	15,8	12,1	9,3	11,8	9,2	12,0	12,4
8 Retired persons (and other inactive aged 65+)	15,8	10,0	11,9	3,6	3,4	14,3	7,1	11,7	5,1	9,8	4,3	10,5	13,3	0,0
9 Other non-employed persons aged < 65	13,9	21,9	27,0	14,8	22,2	14,8	23,5	15,6	20,6	32,3	34,8	16,1	17,8	39,1
<i>Total N</i>	<i>4885</i>	<i>5508</i>	<i>5431</i>	<i>9585</i>	<i>2404</i>	<i>7969</i>	<i>6002</i>	<i>3083</i>	<i>3277</i>	<i>5343</i>	<i>15766</i>	<i>3587</i>	<i>12464</i>	<i>7366</i>
	IT	LT	LU	LV	NL	NO	PL	PT	RO	RS	SE	SI	SK	UK
1 Managers	3,2	6,7	10,0	6,2	7,2	7,5	3,7	4,4	1,5	2,9	5,4	4,3	4,6	8,9
2 Professionals	12,3	18,4	16,2	13,4	20,5	18,9	11,3	9,1	10,2	8,7	23,2	14,8	11,8	11,3
3 Technicians and associated professional employees	9,9	6,3	11,7	9,2	13,2	21,0	5,1	6,1	3,7	5,2	12,4	10,8	14,5	7,5
4 Small entrepreneurs	9,7	5,9	2,5	4,3	5,9	3,5	12,0	7,3	16,5	3,0	5,9	4,4	9,8	5,2
5 Clerks and skilled service employees	11,9	3,7	24,7	5,6	11,4	7,2	5,3	9,4	5,4	5,1	12,2	7,4	9,6	14,2
6 Skilled industrial employees	8,8	13,7	4,8	14,7	7,5	10,6	14,2	13,1	17,0	7,3	11,6	13,2	15,7	7,2
7 Lower status employees	9,3	10,5	5,9	14,0	7,9	13,3	9,4	15,5	8,8	7,9	9,1	11,8	9,9	9,6
8 Retired persons (and other inactive aged 65+)	10,1	8,7	11,3	8,1	3,8	3,2	13,9	9,2	19,8	23,6	2,9	16,6	10,5	10,6
9 Other non-employed persons aged < 65	24,7	26,2	12,9	24,6	22,8	14,7	25,1	25,9	17,0	36,3	17,2	16,7	13,7	25,5
<i>Total N</i>	<i>8678</i>	<i>4249</i>	<i>2961</i>	<i>5021</i>	<i>2985</i>	<i>2655</i>	<i>22460</i>	<i>11306</i>	<i>11011</i>	<i>4414</i>	<i>3070</i>	<i>4013</i>	<i>4255</i>	<i>2795</i>

Source: EU\_AES\_2011\_MICRODATA\_V1.0.SAV, own calculation, not weighted.

Table 3: Level 2: AES 2011 ESEGwa\_2 (working age: 25-64)

	Frequencies	%	Valid %
11 Higher managerial self-employed	1336	0,6	0,7
12 Lower managerial self-employed	1207	0,5	0,7
13 Higher managerial employees	4765	2,1	2,6
14 Lower managerial employees	757	0,3	0,4
21 Science, engineering and information and communications technology (ICT) professionals	5230	2,3	2,9
22 Health professionals	3403	1,5	1,9
23 Business and administration professionals	4387	1,9	2,4
24 Legal, social and cultural professionals	3237	1,4	1,8
25 Teaching professionals	7575	3,4	4,1
31 Science, engineering and ICT technicians and associated professionals	4551	2	2,5
32 Health associate professionals	2997	1,3	1,6
33 Business and administration associate professionals	7107	3,2	3,9
34 Legal, social and cultural associate professionals	2021	0,9	1,1
35 Non-commissioned armed forces officers	187	0,1	0,1
41 Skilled agricultural self-employed workers	5059	2,2	2,8
42 Technicians, clerical support, services and sales self-employed workers	5240	2,3	2,9
43 Craft and related trades self-employed workers	4586	2	2,5
51 General and numerical clerks and other clerical support employees	8227	3,7	4,5
52 Customer services clerks	2066	0,9	1,1
53 Personal care employees	2725	1,2	1,5
54 Armed forced occupations and protective service employees	2373	1,1	1,3
61 Building and related trade employees	3807	1,7	2,1
62 Food processing, wood working, garment employees	2902	1,3	1,6
63 Metal, machinery, handicraft, printing, electrical and electronic trades employees	6590	2,9	3,6
64 Stationary plant and machine operators and assemblers	3976	1,8	2,2
65 Drivers	4914	2,2	2,7
71 Personal services and sales employees	10411	4,6	5,7
72 Blue collar employees and food preparation assistants in elementary occupations	4672	2,1	2,6
73 Cleaners and helpers and services employees in elementary occupations	4037	1,8	2,2
74 Agricultural employees	998	0,4	0,5
80 Retired (and other inactive aged 65 or more)	18794	8,3	10,3
91 Students	2067	0,9	1,1
92 Permanently disabled	5674	2,5	3,1
93 Unemployed not elsewhere classified	18954	8,4	10,4
94 Other inactive aged less than 65 years	13620	6	7,5
95 HU only: inactive	2091	0,9	1,1
<i>Valid Total</i>	<i>182543</i>	<i>81</i>	<i>100</i>




---

MISSING		
99 Age <25 or age >64	24925	11,1
System MISSING	17879	7,9
<i>Total</i>	<i>225347</i>	<i>100</i>

---

Source: EU\_AES\_2011\_MICRODATA\_V1.0.SAV, own calculation, not weighted.

Figure 1: ESeG AES 2011 and ESeG EU-SILC 2011 by country (age: 25-64, only currently employed)



Source: UDB\_c11P\_ver 2011-5 from 01-03-15.SAV and EU\_AES\_2011\_MICRODATA\_V1.0.SAV, own calculation, not weighted. The calculation for EU-SILC is based on the ESeG Tool by Anika Herter and Heike Wirth (2016).

## 5 References

---

- Erikson, Robert and John H. Goldthorpe (1992): *The Constant Flux. A Study of Class. Mobility in Industrial Societies*. Oxford: Clarendon Press.
- European Commission (2013): Commission Regulation (EU) No 912/2013 of 23 September 2013 implementing Regulation (EC) No 452/2008 of the European Parliament and of the Council concerning the production and development of statistics on education and lifelong learning, as regards statistics on education and training systems Text with EEA relevance.
- Herter, Anika and Heike Wirth (2016): SPSS Command Syntax File to generate European Socio-Economic Groups (ESEG 2008) for EU-SILC Data (2011\_eseg08.sps).  
[[https://www.gesis.org/fileadmin/upload/dienstleistung/daten/aml\\_mikrodaten/europ\\_microdata/EU-SILC/Tools/ESeC\\_ESeG\\_Tools/ESeG/CROSS/SPSS/2011\\_eseg08.sps](https://www.gesis.org/fileadmin/upload/dienstleistung/daten/aml_mikrodaten/europ_microdata/EU-SILC/Tools/ESeC_ESeG_Tools/ESeG/CROSS/SPSS/2011_eseg08.sps)]
- Meron Monique and ESSnet members (2014): ESSnet ESeG Final Report. Paris, INSEE, Direction des Statistiques Démographiques et Sociales ESSnet project.  
[[https://ec.europa.eu/eurostat/cros/system/files/ESEG\\_finalReport\\_Vcor30juillet.pdf](https://ec.europa.eu/eurostat/cros/system/files/ESEG_finalReport_Vcor30juillet.pdf)]

## 6 Annex: SPSS routine to generate ESEG using AES 2011

```

* Encoding: windows-1252.
* ESeG_AES2011.sps
* SPSS Command Syntax File
*
* This syntax generates European Socio-Economic Groups (ESEG) for AES 2011 Data
* ESEG_1 (groups)
* ESEGwa_1 (groups, working age 25 to 64)
* ESEG_2 (subgroups)
* ESEGwa_2 (subgroups, working age 25 to 64)
*
* For details on ESeG please see
* Meron, Monique et al. (2014):
* Final Report of the ESSnet on the harmonisation and implementation of a European socio-economic classification:
* European Socio-economic Groups (ESeG)
* Link: https://ec.europa.eu/eurostat/cros/content/eseg\_en
*
* AES 2011 data:
* EU_AES_2011_MICRODATA_V1.0.csv
*
* For transforming the AES 2011 CSV-data (as released by Eurostat) into SPSS-data (*.sav),
* please use the corresponding SPSS Command Syntax File published at
* https://www.gesis.org/en/missy/materials/AES/setups
*
* (c) GESIS 07/06/2018
* GESIS - Leibniz Institute for the Social Sciences
* German Microdata Lab
* Jeanette Bohr
* http://www.gesis.org/en/institute/
*
* Contact: jeanette.bohr@gesis.org
*
* -----
* CONFIGURATION SECTION - Start

* When using SPSS earlier than V16 you might not use "set unicode".
set unicode no.

* The following command should contain the complete path and
* name of the SPSS file, usual file extension "sav";
* Change SPSS_FILENAME to your filename .

GET FILE='SPSS_FILENAME'.

* CONFIGURATION SECTION - End
* There should be probably nothing to change below this line .
* ----- .

set decimal dot.

*****
* Good to know
*
* There are country specific anonymisation rules in AES for MT and HU:
* MT: ISCO08 is provided only for major groups (1-digit)
* HU: MAINSTAT categories 31 to 36 are grouped into a single category (30).
* There are implausible values in MAINSTAT for IE.
*
* => MT, IE: ESeG is not generated for Malta and Ireland.
* => HU: inactive persons are classified in ESeG category 9 (inactive and under 65 years).
*
* The information on occupation (ISCO 08) is only available for currently employed persons.
*
* Non-active persons are split into two groups:

```

\* Group 8: Retired people and non-employed people aged 65 or more

\* Group 9: Non-active persons aged less than 65 years .

\*

\*\*\*\*\*

\*\*\*\*\*

\*

\* Variables used:

\* COUNTRY: Country (alphanumeric)

\* JOBISCO: Occupation ISCO-08 (2-digit) (alphanumeric)

\* JOBSTAT: Status in employment => family workers are classified as self-employed

\* MAINSTAT: Main current labour STATUS

\* AGE: Age

\*

\*\*\*\*\*

\*\*\*\*\*

\* Data Preparation.

\* undo missing value defintion.

MISSING VALUE JOBISCO () MAINSTAT () JOBSTAT ().

\* recode JOBSTAT into numeric variable.

recode JOBISCO

("OC1" = 10)

("OC11" = 11)

("OC12" = 12)

("OC13" = 13)

("OC14" = 14)

("OC2" = 20)

("OC21" = 21)

("OC22" = 22)

("OC23" = 23)

("OC24" = 24)

("OC25" = 25)

("OC26" = 26)

("OC3" = 30)

("OC31" = 31)

("OC32" = 32)

("OC33" = 33)

("OC34" = 34)

("OC35" = 35)

("OC4" = 40)

("OC41" = 41)

("OC42" = 42)

("OC43" = 43)

("OC44" = 44)

("OC49" = 49)

("OC5" = 50)

("OC51" = 51)

("OC52" = 52)

("OC53" = 53)

("OC54" = 54)

("OC6" = 60)

("OC61" = 61)

("OC62" = 62)

("OC63" = 63)

("OC7" = 70)

("OC71" = 71)

("OC72" = 72)

("OC73" = 73)

("OC74" = 74)

("OC75" = 75)

("OC8" = 80)

("OC81" = 81)

("OC82" = 82)

("OC83" = 83)

("OC9" = 90)

("OC91" = 91)  
 ("OC92" = 92)  
 ("OC93" = 93)  
 ("OC94" = 94)  
 ("OC95" = 95)  
 ("OC96" = 96)  
 ("OC0" = 0)  
 ("OC01" = 01)  
 ("OC02" = 02)  
 ("OC03" = 03)  
 ("-1" = -1)  
 ("-2" = -2)  
 into ISCO08.

val lab ISCO08

10 "Managers"  
 11 "Chief executives, senior officials and legislators"  
 12 "Administrative and commercial managers"  
 13 "Production and specialised services managers"  
 14 "Hospitality, retail and other services managers"  
 20 "Professionals"  
 21 "Science and engineering professionals"  
 22 "Health professionals"  
 23 "Teaching professionals"  
 24 "Business and administration professionals"  
 25 "Information and communications technology professionals"  
 26 "Legal, social and cultural professionals"  
 30 "Technicians and associate professionals"  
 31 "Science and engineering associate professionals"  
 32 "Health associate professionals"  
 33 "Business and administration associate professionals"  
 34 "Legal, social, cultural and related associate professionals"  
 35 "Information and communications technicians"  
 40 "Clerical support workers"  
 41 "General and keyboard clerks"  
 42 "Customer services clerks"  
 43 "Numerical and material recording clerks"  
 44 "Other clerical support workers"  
 50 "Service and sales workers"  
 51 "Personal service workers"  
 52 "Sales workers"  
 53 "Personal care workers"  
 54 "Protective services workers"  
 60 "Skilled agricultural, forestry and fishery workers"  
 61 "Market-oriented skilled agricultural workers"  
 62 "Market-oriented skilled forestry, fishery and hunting workers"  
 63 "Subsistence farmers, fishers, hunters and gatherers"  
 70 "Craft and related trades workers"  
 71 "Building and related trades workers, excluding electricians"  
 72 "Metal, machinery and related trades workers"  
 73 "Handicraft and printing workers"  
 74 "Electrical and electronic trades workers"  
 75 "Food processing, wood working, garment and other craft and related trades workers"  
 80 "Plant and machine operators, and assemblers"  
 81 "Stationary plant and machine operators"  
 82 "Assemblers"  
 83 "Drivers and mobile plant operators"  
 90 "Elementary occupations"  
 91 "Cleaners and helpers"  
 92 "Agricultural, forestry and fishery labourers"  
 93 "Labourers in mining, construction, manufacturing and transport"  
 94 "Food preparation assistants"  
 95 "Street and related sales and service workers"  
 96 "Refuse workers and other elementary workers"  
 0 "Armed forces occupations"  
 01 "Commissioned armed forces officers"  
 02 "Non-commissioned armed forces officers"  
 03 "Armed forces occupations, other ranks"  
 -1 "No answer"  
 -2 "Not applicable (MAINSTAT not equal 11,12)".

\* exclude Malta and Irland.

\* MT: ISCO08 is provided only for major groups (1-digit) => no ESEG is generated for Malta.

do if (COUNTRY='MT').

recode ISCO08 (-2 thru 96=sysmis).

end if.

execute.

\* IE: implausible values in MAINSTAT => no ESEG is generated for Ireland.

do if (COUNTRY='IE').

recode ISCO08 (-2 thru 96=sysmis).

end if.

execute.

\* Employment status variable (family workers are defined as self-employed).

compute STA = JOBSTAT.

if (JOBSTAT = 10 or JOBSTAT = 11 or JOBSTAT = 12 or JOBSTAT = 30) STA = 1.

if (JOBSTAT = 21 or JOBSTAT = 22) STA = 2 .

val labs STA 1 'self-employed' 2 'employed'.

\*\*\*\*\*

\* ESEG\_1 = aggregated level (groups).

\* ESEG 1 to 7: Assigning ESEG to respondents with Main current labour STATUS (MAINSTAT): employed (part-time, full-time).

if ((ISCO08 >= 11 and ISCO08 <= 14) and (STA = 1 or STA = 2)) or (ISCO08 = 01 and STA = 2) and ((MAINSTAT >= 11 and MAINSTAT <= 12)) ESEG\_1 = 1 .

if ((ISCO08 >= 21 and ISCO08 <= 26) and (STA = 1 or STA = 2)) and ((MAINSTAT >= 11 and MAINSTAT <= 12)) ESEG\_1 = 2 .

if ((ISCO08 >= 31 and ISCO08 <= 35) or (ISCO08 = 02)) and (STA = 2) and ((MAINSTAT >= 11 and MAINSTAT <= 12)) ESEG\_1 = 3.

if ((ISCO08 >= 31 and ISCO08 <= 96) and STA = 1) and ((MAINSTAT >= 11 and MAINSTAT <= 12)) ESEG\_1 = 4 .

if ((ISCO08 >= 41 and ISCO08 <= 44) or (ISCO08 = 53 or ISCO08 = 54 or ISCO08 = 03)) and (STA = 2) and ((MAINSTAT >= 11 and MAINSTAT <= 12)) ESEG\_1 = 5 .

if ((ISCO08 >= 71 and ISCO08 <= 83) and STA = 2) and ((MAINSTAT >= 11 and MAINSTAT <= 12) ) ESEG\_1 = 6.

if ((ISCO08 = 51 or ISCO08 = 52) or (ISCO08 >= 61 & ISCO08 <= 63) or (ISCO08 >= 91 and ISCO08 <= 96)) and (STA = 2) and ((MAINSTAT >= 11 and MAINSTAT <= 12)) ESEG\_1 = 7.

\* ESEG 8 'Retired persons': Retired or Respondents > 65 and non-employed.

if (MAINSTAT = 32 or (AGE >= 65 and (MAINSTAT = 20 or MAINSTAT = 31 or MAINSTAT = 33 or MAINSTAT = 35 or MAINSTAT = 36))) ESEG\_1 = 8.

\* ESEG 9 ='other non-employed persons': Respondents <65 and non-employed according to MAINSTAT (31,33,35,36)/

\* for Hungary: MAINSTAT (30).

if ((MAINSTAT = 31 or MAINSTAT = 33 or MAINSTAT = 35 or MAINSTAT = 36) and AGE <= 64) ESEG\_1 = 9.

if ((MAINSTAT = 30) and AGE <= 64) ESEG\_1 = 9.

\* ESEG 9 = 'other non-employed persons': unemployed.

if (MAINSTAT = 20) ESEG\_1 = 9.

\* MISSING.

\* ESEG for Malta a is set to missing value, because of COUNTRY specific anonymisation (ISCO08 only available as 1-digit).

do if (COUNTRY = 'MT') .

recode ESEG\_1 (0 thru 99 = sysmis).

end if.

\* ESEG for Ireland a is set to missing value, because of implausible values in MAINSTAT.

do if (COUNTRY = 'IE') .

recode ESEG\_1 (0 thru 99 = sysmis).

end if.

variable label ESEG\_1 "ESeG level 1".

value labels ESEG\_1

1 "Managers"

2 "Professionals"

3 "Technicians and associated professional employees"

4 "Small entrepreneurs"

5 "Clerks and skilled service employees"

6 "Skilled industrial employees"  
 7 "Lower status employees"  
 8 "Retired persons (and other inactive aged 65 or more)"  
 9 "Other non-employed persons aged < 65".  
 frequencies ESEG\_1.

\*\*\*\*\*

\* ESEGwa\_1 = aggregated level (groups), aged 25 to 64 years.

compute ESEGwa\_1= ESEG\_1.  
 do if (age <=24 or age >=65).  
 compute ESEGwa\_1=99.  
 end if.  
 missing value ESEGwa\_1 (99).  
 variable label ESEGwa\_1 "ESeG level 1, working age".  
 value labels ESEGwa\_1  
 1 "Managers"  
 2 "Professionals"  
 3 "Technicians and associated professional employees"  
 4 "Small entrepreneurs"  
 5 "Clerks and skilled service employees"  
 6 "Skilled industrial employees"  
 7 "Lower status employees"  
 8 "Retired persons (and other inactive aged 65 or more)"  
 9 "Other non-employed persons aged < 65"  
 99 "Age <25 or age >64".  
 frequencies ESEGwa\_1.

\*\*\*\*\*

\* ESEG\_2 = detailed level (subgroups).

\* Definition

\*1 Managers  
 \*1.1 Higher managerial self-employed (ISCO 11, 12, 13 and STATUS=se)  
 \*1.2 Lower managerial self-employed (ISCO 14 and STATUS=se)  
 \*1.3 Higher managerial employees (ISCO 11, 12, 13 and 01 and STATUS=e)  
 \*1.4 Lower managerial employees (ISCO 14 and STATUS=e)  
 \*/.

if ((ISCO08 = 11 or ISCO08 = 12 or ISCO08 = 13) and STA = 1 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 11 .  
 if (ISCO08 = 14 and STA = 1 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 12 .  
 if ((ISCO08 = 01 or ISCO08 = 11 or ISCO08 = 12 or ISCO08 = 13) and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 13.  
 if (ISCO08 = 14 and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 14.

\* Definition

\*2 Professionals  
 \*2.1 Science, engineering and information and communications technology (ICT) professionals (ISCO 21,25)  
 \*2.2 Health professionals (ISCO 22)  
 \*2.3 Business and administration professionals (ISCO 24)  
 \*2.4 Legal, social and cultural professionals (ISCO 26)  
 \*2.5 Teaching professionals (ISCO 23)  
 \*/.

if ((ISCO08 = 21 or ISCO08 = 25) and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 21.  
 if (ISCO08 = 22 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 22.  
 if (ISCO08 = 24 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 23.  
 if (ISCO08 = 26 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 24.  
 if (ISCO08 = 23 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 25.

\* Definition

\*3 Technicians and associated professionals employees (STATUS= e)  
 \*3.1 Science, engineering and ICT technicians and associated professionals (ISCO 31, 35)  
 \*3.2 Health associate professionals (ISCO 32)  
 \*3.3 Business and administration associate professionals (ISCO 33)  
 \*3.4 Legal, social and cultural associate professionals (ISCO 34)



\*3.5 Non-commissioned armed forces officers (ISCO 02 )

\*/.

if ((ISCO08 = 31 or ISCO08 = 35) and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 31.

if (ISCO08 = 32 and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 32.

if (ISCO08 = 33 and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 33.

if (ISCO08 = 34 and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 34.

if (ISCO08 = 02 and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 35.

\* Definition

\*4 Small entrepreneurs (STATUS= se)

\*4.1 Skilled agricultural self employed workers (ISCO 6)

\*4.2 Technicians, clerical support, services and sales self employed workers (ISCO 3, 4, 5)

\*4.3 Craft and related trades self employed workers (ISCO 7, 8, 9)

\*/.

if (ISCO08 >= 61 and ISCO08 <= 63 and STA= 1 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 41.

if (ISCO08 >= 31 and ISCO08 <= 54 and STA= 1 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 42.

if (ISCO08 >= 71 and ISCO08 <= 96 and STA= 1 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 43.

\* Definition

\*5 Clerks and skilled service workers (STATUS=e)

\*5.1 General and numerical clerks and other clerical support employees (ISCO 41, 43, 44)

\*5.2 Customer services clerks (ISCO 42)

\*5.3 Personal care employees (ISCO 53)

\*5.4 Armed forced occupations and protective service employees (ISCO 03 and ISCO 54)

\*/.

if ((ISCO08 = 41 or ISCO08 = 43 or ISCO08 = 44) and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 51.

if (ISCO08 = 42 and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 52.

if (ISCO08 = 53 and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 53.

if ((ISCO08 = 03 or ISCO08 = 54) and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 54.

\* Definition

\*6 Industrial and agricultural employees (STATUS= e)

\*6.1 Building and related trade employees (ISCO 71)

\*6.2 Food processing, wood working, garment employees (ISCO 75)

\*6.3 Metal, machinery, handicraft, printing, electrical and electronic trades employees (ISCO 72, 73, 74)

\*6.4 Stationary plant and machine operators and assemblers (ISCO 81, 82)

\*6.5 Drivers (ISCO 83)

\*/.

if (ISCO08 = 71 and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 61.

if (ISCO08 = 75 and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 62.

if ((ISCO08 = 72 or ISCO08 = 73 or ISCO08 = 74) and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 63.

if ((ISCO08 = 81 or ISCO08 = 82) and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 64.

if (ISCO08 = 83 and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 65.

\* Definition

\*7 Less skilled workers (STATUS= e)

\*7.1 Personal services and sales employees (ISCO 51, 52)

\*7.2 Blue collar employees and food preparation assistants in elementary occupations (ISCO 92, 93, 94, 96)

\*7.3 Cleaners and helpers and services employees in elementary occupations (ISCO 91, 95)

\*7.4 Agricultural employees (ISCO 6)

\*/.

if ((ISCO08 = 51 or ISCO08 = 52) and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 71.

if ((ISCO08 = 92 or ISCO08 = 93 or ISCO08 = 94 or ISCO08 = 96) and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 72.

if ((ISCO08 = 91 or ISCO08 = 95) and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 73.

if ((ISCO08 >= 61 and ISCO08 <= 63) and STA = 2 and ESEG\_1 > 0 and ESEG\_1 < 8) ESEG\_2 = 74.

\* Definition

\*8 Retired persons (and persons not employed 65 years or more)

\*8.1 Retired Managers

\*8.2 Retired professionals

\*8.3 Retired technicians and associate professionals

\*8.4 Retired small entrepreneurs

\*8.5 Retired skilled service workers

\*8.6 Retired skilled blue-collars

\*8.7 Retired less skilled workers

\*8.8 Other inactive aged 65 or more

\*definition of 8.1 thru 8.8. not possible (jobisco only available for MAINSTAT=11 and MAINSTAT=12).\*

\*/.

if (ESEG\_1 = 8) ESEG\_2 = 80.

\* Definition

\*9 Other non employed persons

\*9.1 Student

\*9.2 Permanently disabled

\*9.3 Unemployed not elsewhere classified

\*9.4 Other inactive aged less than 65 years

\*9.5 HU only: inactive (MAINSTAT 31 to 36)

\*/.

if (MAINSTAT = 31 and ESEG\_1 = 9) ESEG\_2 = 91.

if (MAINSTAT = 33 and ESEG\_1 = 9) ESEG\_2 = 92.

if (MAINSTAT = 20 and ESEG\_1 = 9) ESEG\_2 = 93.

if ((MAINSTAT = 34 or MAINSTAT = 35 or MAINSTAT = 36) and ESEG\_1 = 9) ESEG\_2 = 94.

if (MAINSTAT = 30 and ESEG\_1=9) ESEG\_2 = 95.

\* MISSING.

do if (missing(ESEG\_1)) .

recode ESEG\_2 (0 thru 99 = sysmis).

end if.

variable label ESEG\_2 "ESeG level 2".

value labels ESEG\_2

11 "Higher managerial self-employed"

12 "Lower managerial self-employed"

13 "Higher managerial employees"

14 "Lower managerial employees"

21 "Science, engineering and information and communications technology (ICT) professionals"

22 "Health professionals"

23 "Business and administration professionals"

24 "Legal, social and cultural professionals"

25 "Teaching professionals"

31 "Science, engineering and ICT technicians and associated professionals"

32 "Health associate professionals"

33 "Business and administration associate professionals"

34 "Legal, social and cultural associate professionals"

35 "Non-commissioned armed forces officers"

41 "Skilled agricultural self-employed workers"

42 "Technicians, clerical support, services and sales self-employed workers"

43 "Craft and related trades self-employed workers"

51 "General and numerical clerks and other clerical support employees"

52 "Customer services clerks"

53 "Personal care employees"

54 "Armed forced occupations and protective service employees"

61 "Building and related trade employees"

62 "Food processing, wood working, garment employees"

63 "Metal, machinery, handicraft, printing, electrical and electronic trades employees"

64 "Stationary plant and machine operators and assemblers"

65 "Drivers"

71 "Personal services and sales employees"

72 "Blue collar employees and food preparation assistants in elementary occupations"

73 "Cleaners and helpers and services employees in elementary occupations"

74 "Agricultural employees"

80 "Retired (and other inactive aged 65 or more)"

91 "Students"

92 "Permanently disabled"

93 "Unemployed not elsewhere classified"

94 "Other inactive aged less than 65 years"

95 "HU only: inactive".

frequencies ESEG\_2.

```

*****
* ESEGwa_2 = aggregated level (subgroups), aged 25 to 64 years.

compute ESEGwa_2= ESEG_2.
do if (age <=24 or age >=65).
compute ESEGwa_2=99.
end if.
missing value ESEGwa_2 (99).
variable label ESEGwa_2 "ESeG level 2, working age".
value labels ESEGwa_2
11 "Higher managerial self-employed"
12 "Lower managerial self-employed"
13 "Higher managerial employees"
14 "Lower managerial employees"
21 "Science, engineering and information and communications technology (ICT) professionals"
22 "Health professionals"
23 "Business and administration professionals"
24 "Legal, social and cultural professionals"
25 "Teaching professionals"
31 "Science, engineering and ICT technicians and associated professionals"
32 "Health associate professionals"
33 "Business and administration associate professionals"
34 "Legal, social and cultural associate professionals"
35 "Non-commissioned armed forces officers"
41 "Skilled agricultural self-employed workers"
42 "Technicians, clerical support, services and sales self-employed workers"
43 "Craft and related trades self-employed workers"
51 "General and numerical clerks and other clerical support employees"
52 "Customer services clerks"
53 "Personal care employees"
54 "Armed forced occupations and protective service employees"
61 "Building and related trade employees"
62 "Food processing, wood working, garment employees"
63 "Metal, machinery, handicraft, printing, electrical and electronic trades employees"
64 "Stationary plant and machine operators and assemblers"
65 "Drivers"
71 "Personal services and sales employees"
72 "Blue collar employees and food preparation assistants in elementary occupations"
73 "Cleaners and helpers and services employees in elementary occupations"
74 "Agricultural employees"
80 "Retired (and other inactive aged 65 or more)"
91 "Students"
92 "Permanently disabled"
93 "Unemployed not elsewhere classified"
94 "Other inactive aged less than 65 years"
95 "HU only: inactive"
99 "Age <25 or age >64".
frequencies ESEGwa_2.

```